

Uno studio per la Befana 2015 e rifatto nel 2016...  
( mancano 84 anni al 2100 )

## Lorentz in modo semplicissimo

### *1 Prima variante:*

Questa funzione crea la matrice di Lorentz facendo pero' uso della funzione sqrt(...) quindi con un piccolo rischio di complicazioni numeriche ossia che Maxima si incasini e non riesca a semplificare le espressioni.

---

Questa funzione ha bisogno di due dati:

1) Il vettore che specifica la direzione della trasformazione ossia un vettore di 4 elementi di cui almeno uno dei primi tre elementi non deve essere nullo e l'ultimo, che controlla la formula globale, DEVE ESSERE OBBLIGATORIAMENTE POSITIVO, MAI NULLO.

2) La velocita' della luce ossia un numero positivo e mai nullo.

---

Questa funzione produce tre risultati ossia :

1) La trasformazione di Lorentz.

2) Il quadrivettore che viene "fermato" dalla trasformazione di Lorentz ossia la trasformazione di Lorentz moltiplicata per questo vettore da' un vettore con le componenti spaziali tutte nulle.

3) Il tensore metrico covariante che si usa e che dipende da quale e' considerata la velocita' della luce nel vuoto.

---

```
(%i1) lsemplice(br, lux) := block([bd, bx, by, bz, cm, sh, lz, uu, gm],
  bd:sqrt(br[1]*br[1]+br[2]*br[2]+br[3]*br[3]),
  bx:br[1]/bd,
  by:br[2]/bd,
  bz:br[3]/bd,
  cm:(br[4]+1/br[4])/2-1,
  sh:(br[4]-1/br[4])/2,
  lz:matrix([1+cm, -sh*bx/lux, -sh*by/lux, -sh*bz/lux],
  [-lux*sh*bx, 1+cm*bx*bx, cm*bx*by, cm*bx*bz],
  [-lux*sh*by, cm*bx*by, 1+cm*by*by, cm*by*bz],
  [-lux*sh*bz, cm*bx*bz, cm*by*bz, 1+cm*bz*bz]),
  uu:[1+cm, lux*sh*bx, lux*sh*by, lux*sh*bz],
  gm:matrix([lux*lux, 0, 0, 0],
  [0, -1, 0, 0], [0, 0, -1, 0], [0, 0, 0, -1]),
  return ([lz, uu, gm]))$
```

Verifico che la funzione data funzioni bene scegliendo bene i dati in modo da non ottenere radici quadrate. Quindi Maxima non deve fare praticamente nessuna fatica a semplificare i risultati...  
Definisco tre vettori il cui significato sara' spiegato poi...

```
(%i2) cluce:7;
      besea:[3/13,4/13,12/13,9];
      beseaq:[3/13,4/13,12/13,81];
      beseasq:[3/13,4/13,12/13,3];
```

```
(%o2) 7
```

```
(%o3) [ $\frac{3}{13}$ ,  $\frac{4}{13}$ ,  $\frac{12}{13}$ , 9]
```

```
(%o4) [ $\frac{3}{13}$ ,  $\frac{4}{13}$ ,  $\frac{12}{13}$ , 81]
```

```
(%o5) [ $\frac{3}{13}$ ,  $\frac{4}{13}$ ,  $\frac{12}{13}$ , 3]
```

Ecco il calcolo applicato al primo vettore....

```
(%i6) lzz:lsemplice(besea,cluce);
```

```
(%o6) [ $\begin{bmatrix} \frac{41}{9} & -\frac{40}{273} & \frac{160}{819} & \frac{160}{273} \\ -\frac{280}{39} & \frac{201}{169} & \frac{128}{507} & \frac{128}{169} \\ \frac{1120}{117} & \frac{128}{507} & \frac{2033}{1521} & \frac{512}{507} \\ -\frac{1120}{39} & \frac{128}{169} & \frac{512}{507} & \frac{681}{169} \end{bmatrix}$ , [ $\frac{41}{9}$ ,  $\frac{280}{39}$ ,  $\frac{1120}{117}$ ,  $\frac{1120}{39}$ ], [ $\begin{bmatrix} 49 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix}$ ]
```

Verifico che la trasformazione di Lorentz, applicata al quadrivettore calcolato, lo fermi veramente ed inoltre la trasformazione di Lorentz, applicata al tensore metrico covariante, non lo modifica ossia lo lascia immutato...

---

Attenzione. Se la velocita' della luce non vale 1, la matrice della trasformazione di Lorentz NON E' SIMMETRICA per cui va usata anche la sua trasposta ottenuta con la transpose(...) di Maxima.

```
(%i7) print("La trasformata di Lorentz e' lzz[1] ",lzz[1])$
print("lzz[2] e' il vettore usato= ",lzz[2])$
print("La norma di lzz[2]= ",lzz[2].lzz[3].lzz[2])$
print("Usando lzz[1] il vettore dovrebbe essere ",
      "FERMO ossia zero tutte le sue componenti spaziali")$
unuovo:lzz[1].lzz[2]$
print("unuovo = ",unuovo)$
print("La norma di unuovo deve essere invariata ",
      "ossia unuovo.lzz[3].unuovo = ",
      unuovo.lzz[3].unuovo)$
print("Come e' il tensore metrico all'inizio =",
      lzz[3])$
print("Dopo la trasformazione il tens.metrico",
      " deve essere identico")$
transpose(lzz[1]).lzz[3].lzz[1];
print("Ho scritto: transpose(lzz[1]).lzz[3].lzz[1] ")$
```

La trasformata di Lorentz e' lzz[1]

$$\begin{bmatrix} \frac{41}{9} & -\frac{40}{273} & \frac{160}{819} & \frac{160}{273} \\ -\frac{280}{39} & \frac{201}{169} & \frac{128}{507} & \frac{128}{169} \\ \frac{1120}{117} & \frac{128}{507} & \frac{2033}{1521} & \frac{512}{507} \\ -\frac{1120}{39} & \frac{128}{169} & \frac{512}{507} & \frac{681}{169} \end{bmatrix}$$

lzz[2] e' il vettore usato=  $[\frac{41}{9}, \frac{280}{39}, \frac{1120}{117}, \frac{1120}{39}]$

La norma di lzz[2]= 49

Usando lzz[1] il vettore dovrebbe essere

FERMO ossia zero tutte le sue componenti spaziali

unuovo =  $\begin{bmatrix} 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}$

La norma di unuovo deve essere invariata ossia unuovo.lzz[3].unuovo = 49

Come e' il tensore metrico all'inizio =  $\begin{bmatrix} 49 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix}$

Dopo la trasformazione il tens.metrico deve essere identico

(%o16)  $\begin{bmatrix} 49 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix}$

Ho scritto: transpose(lzz[1]).lzz[3].lzz[1]

Qui mostro come fare il quadrato della trasformazione di Lorentz ossia cosa viene, applicata due volte di seguito...

```
(%i18) lzzq:lsemplce(beseaq,cluce);
```

$$\left( \%o18 \right) \left[ \begin{array}{cccc} \frac{3281}{81} & \frac{3280}{2457} & \frac{13120}{7371} & \frac{13120}{2457} \\ \frac{22960}{351} & \frac{4721}{1521} & \frac{12800}{4563} & \frac{12800}{1521} \\ \frac{91840}{1053} & \frac{12800}{4563} & \frac{64889}{13689} & \frac{51200}{4563} \\ \frac{91840}{351} & \frac{12800}{1521} & \frac{51200}{4563} & \frac{52721}{1521} \end{array} \right], \left[ \frac{3281}{81}, \frac{22960}{351}, \frac{91840}{1053}, \frac{91840}{351} \right], \left[ \begin{array}{cccc} 49 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{array} \right]$$

Dimostro che effettivamente lzzq[1] e' il quadrato di lzz[1].

```
(%i19) lzzq[1]-lzz[1].lzz[1];
```

```
(%o19) [ 0 0 0 0
         0 0 0 0
         0 0 0 0
         0 0 0 0 ]
```

Dimostro come si fa la radice quadrata della matrice che rappresenta la trasformazione di Lorentz ossia lzzsq[1] moltiplicata per se stessa deve dare lzz[1]...

```
(%i20) lzzsq:lsemplce(beseasq,cluce);
```

```
(%o20) [ [ 5 4 16 16
           3 91 273 91
           28 175 8 24
           13 169 169 169
           112 8 539 32
           39 169 507 169
           112 24 32 265
           13 169 169 169 ] , [ 5 28 112 112
                               3 13 39 13 ] , [ 49 0 0 0
                                                0 -1 0 0
                                                0 0 -1 0
                                                0 0 0 -1 ] ]
```

Dimostro che realmente lzzsq[1] e' la radice quadrata di lzz[1]...

```
(%i21) lzz[1]-lzzsq[1].lzzsq[1];
```

```
(%o21) [ 0 0 0 0
         0 0 0 0
         0 0 0 0
         0 0 0 0 ]
```

□ **2 Seconda variante:  
W la razionalita' !!!**

Questa funzione e' simile a quella precedente ma ha il GRANDE PREGIO DI NON FARE USO DELLA sqrt(...). Per evitare problemi impedisco di usare come primo argomento il valore zero che darebbe divisione per zero e dunque dati infiniti ossia errore di calcolo.... Il secondo ed il terzo argomento possono essere dati qualsiasi ed il quarto e' la velocita' della luce nel vuoto che deve ovviamente non essere mai uno zero.. Se si assegna zero viene usato il valore 1.

```
(%i22) lrazionale(rd,f,h,xlux):=block([r,lux,bx,by,bz,cm,sh,
    lz,uu,gm],
    if rd=0 then r:1 else r:rd,
    if xlux=0 then lux:1 else lux:abs(xlux),
    bx:2*h*(1-f*f)/((1+f*f)*(1+h*h)),
    by:4*h*f/((1+f*f)*(1+h*h)),
    bz:(1-h*h)/(1+h*h),
    cm:(r+1/r)/2-1,
    sh:(r-1/r)/2,
    lz:matrix([1+cm,-sh*bx/lux,-sh*by/lux,-sh*bz/lux],
    [-lux*sh*bx,1+cm*bx*bx,cm*bx*by,cm*bx*bz],
    [-lux*sh*by,cm*bx*by,1+cm*by*by,cm*by*bz],
    [-lux*sh*bz,cm*bx*bz,cm*by*bz,1+cm*bz*bz]),
    uu:[1+cm,lux*sh*bx,lux*sh*by,lux*sh*bz],
    gm:matrix([lux*lux,0,0,0],
    [0,-1,0,0],[0,0,-1,0],[0,0,0,-1]),
    return ([lz,uu,gm]))$
```

Ecco l'applicazione della funzione che produce sempre numeri razionali se gli argomenti di ingresso sono anche loro razionali...

```
(%i23) lrr:lrazionale(1+random(7)/11,random(19)/17,
    random(23)/19,cluce);
lrr[2].lrr[3].lrr[2];
urnuovo:lrr[1].lrr[2];
urnuovo.lrr[3].urnuovo;
print("Come e' il tensore metrico all'inizio")$
lrr[3];
print("Dopo la trasformazione ",
    " il tens.metrico deve essere identico")$
transpose(lrr[1]).lrr[3].lrr[1];
```

$$\begin{matrix}
 (%o23) & \left[ \begin{array}{cccc}
 \frac{265}{264} & \frac{41515}{8505497} & \frac{29716}{25516491} & \frac{2645}{232232} \\
 \frac{290605}{1215071} & \frac{134296150331}{134217957731} & \frac{18656480}{134217957731} & \frac{622725}{458081767} \\
 \frac{208012}{3645213} & \frac{18656480}{134217957731} & \frac{402667227305}{402653873193} & \frac{148580}{458081767} \\
 \frac{18515}{33176} & \frac{622725}{458081767} & \frac{148580}{458081767} & \frac{12547027}{12507352}
 \end{array} \right], & \left[ \frac{265}{264}, \frac{290605}{1215071}, \frac{208012}{3645213}, \frac{18515}{33176} \right],
 \end{matrix}$$

$$\begin{bmatrix}
 49 & 0 & 0 & 0 \\
 0 & -1 & 0 & 0 \\
 0 & 0 & -1 & 0 \\
 0 & 0 & 0 & -1
 \end{bmatrix}$$

```
(%o24) 49
```

$$\begin{bmatrix}
 1 \\
 0 \\
 0 \\
 0
 \end{bmatrix}$$

```
(%o26) 49
```

Come e' il tensore metrico all'inizio

$$(%o28) \begin{bmatrix}
 49 & 0 & 0 & 0 \\
 0 & -1 & 0 & 0 \\
 0 & 0 & -1 & 0 \\
 0 & 0 & 0 & -1
 \end{bmatrix}$$

Dopo la trasformazione il tens.metrico deve essere identico

$$(%o30) \begin{bmatrix}
 49 & 0 & 0 & 0 \\
 0 & -1 & 0 & 0 \\
 0 & 0 & -1 & 0 \\
 0 & 0 & 0 & -1
 \end{bmatrix}$$

```
(%i31) lrra:lrazionale(25,4,7,cluce);
lrraq:lrazionale(25*25,4,7,cluce);
lrrasq:lrazionale(5,4,7,cluce);
```

$$\begin{matrix}
 (%o31) & \left[ \begin{array}{cccc}
 \frac{313}{25} & \frac{936}{2125} & \frac{2496}{10625} & \frac{7488}{4375} \\
 \frac{45864}{2125} & \frac{307633}{180625} & \frac{338688}{903125} & \frac{145152}{53125} \\
 \frac{122304}{10625} & \frac{338688}{903125} & \frac{5418793}{4515625} & \frac{387072}{265625} \\
 \frac{52416}{625} & \frac{145152}{53125} & \frac{387072}{265625} & \frac{181513}{15625}
 \end{array} \right], & \left[ \frac{313}{25}, -\frac{45864}{2125}, \frac{122304}{10625}, -\frac{52416}{625} \right], \\
 & \left[ \begin{array}{cccc}
 49 & 0 & 0 & 0 \\
 0 & -1 & 0 & 0 \\
 0 & 0 & -1 & 0 \\
 0 & 0 & 0 & -1
 \end{array} \right] \\
 & \left[ \begin{array}{cccc}
 \frac{195313}{625} & \frac{585936}{53125} & \frac{1562496}{265625} & \frac{4687488}{109375} \\
 \frac{28710864}{53125} & \frac{90373033}{4515625} & \frac{228953088}{22578125} & \frac{98122752}{1328125} \\
 \frac{76562304}{265625} & \frac{228953088}{22578125} & \frac{723432193}{112890625} & \frac{261660672}{6640625} \\
 \frac{32812416}{15625} & \frac{98122752}{1328125} & \frac{261660672}{6640625} & \frac{112530913}{390625}
 \end{array} \right], & \left[ \frac{195313}{625}, -\frac{28710864}{53125}, \frac{76562304}{265625}, -\frac{32812416}{15625} \right], \\
 & \left[ \begin{array}{cccc}
 49 & 0 & 0 & 0 \\
 0 & -1 & 0 & 0 \\
 0 & 0 & -1 & 0 \\
 0 & 0 & 0 & -1
 \end{array} \right] \\
 & \left[ \begin{array}{cccc}
 \frac{13}{5} & \frac{36}{425} & \frac{96}{2125} & \frac{288}{875} \\
 \frac{1764}{425} & \frac{39653}{36125} & \frac{9408}{180625} & \frac{4032}{10625} \\
 \frac{4704}{2125} & \frac{9408}{180625} & \frac{928213}{903125} & \frac{10752}{53125} \\
 \frac{2016}{125} & \frac{4032}{10625} & \frac{10752}{53125} & \frac{7733}{3125}
 \end{array} \right], & \left[ \frac{13}{5}, -\frac{1764}{425}, \frac{4704}{2125}, -\frac{2016}{125} \right], & \left[ \begin{array}{cccc}
 49 & 0 & 0 & 0 \\
 0 & -1 & 0 & 0 \\
 0 & 0 & -1 & 0 \\
 0 & 0 & 0 & -1
 \end{array} \right]
 \end{matrix}$$

Bene se ottengo solo zeri...

```
(%i34) lrra[1].lrra[1]-lrraq[1];
lrrasq[1].lrrasq[1]-lrra[1];
```

$$\begin{matrix}
 (%o34) & \left[ \begin{array}{cccc}
 0 & 0 & 0 & 0 \\
 0 & 0 & 0 & 0 \\
 0 & 0 & 0 & 0 \\
 0 & 0 & 0 & 0
 \end{array} \right] \\
 (%o35) & \left[ \begin{array}{cccc}
 0 & 0 & 0 & 0 \\
 0 & 0 & 0 & 0 \\
 0 & 0 & 0 & 0 \\
 0 & 0 & 0 & 0
 \end{array} \right]
 \end{matrix}$$

Altre funzioni di servizio, utili per gestire i vettori con le convenzioni usate qui...

Questa piccola funzione normalizza ad 1 la sola parte spaziale di un quadrivettore a meno che non abbia tutte le componenti spaziali nulle nel qual caso da' un vettore di 3 dimensioni di zeri.

```
(%i36) v3norma(vv):=block([vsq,vn],
    vsq:sqrt(vv[2]^2+vv[3]^2+vv[4]^2),
    if vsq=0 then return([0,0,0]),
    vn:ratsimp([vv[2]/vsq,vv[3]/vsq,vv[4]/vsq]),
    return(vn))$
```

In coordinate sferiche indicate da ra, f ed h puo' servire passare alle tradizionali cartesiane...

```
(%i37) incart(ra,f,h):=block([bx,by,bz],
    bx:2*h*(1-f*f)/((1+f*f)*(1+h*h)),
    by:4*h*f/((1+f*f)*(1+h*h)),
    bz:(1-h*h)/(1+h*h),
    return([ra*bx,ra*by,ra*bz]) )$
```

### 3 Conclusione

Le due funzioni qui definite e sperimentate: lsemplice(br,lux) e lrazionale(rd,f,h,xlux) chiariscono la struttura della trasformazione di Lorentz e, la seconda, a mio parere, e' utilissima per lavorare solo con numeri razionali e quindi operando con la massima precisione possibile su qualsiasi calcolatore.

```
(%i38) print("Qui sono definite due funzioni:")$
print("lsemplice(br,lux)")$
print("lrazionale(rd,f,h,xlux)")$
print("E varie variabili, ad esempio: besea =",besea,
    " e cluce=",cluce)$
print("Matrici lzz[1]=",lzz[1]," e lrra[1]=",lrra[1])$
```

Qui sono definite due funzioni:

*lsemplice*(br,lux)

*lrazionale*(rd,f,h,xlux)

E varie variabili, ad esempio: besea =  $[\frac{3}{13}, \frac{4}{13}, \frac{12}{13}, 9]$  e cluce=7

$$\text{Matrici } lzz[1]= \begin{bmatrix} \frac{41}{9} & -\frac{40}{273} & -\frac{160}{819} & -\frac{160}{273} \\ -\frac{280}{39} & \frac{201}{169} & \frac{128}{507} & \frac{128}{169} \\ -\frac{1120}{117} & \frac{128}{507} & \frac{2033}{1521} & \frac{512}{507} \\ -\frac{1120}{39} & \frac{128}{169} & \frac{512}{507} & \frac{681}{169} \end{bmatrix} \quad \text{e } lrra[1]= \begin{bmatrix} \frac{313}{25} & \frac{936}{2125} & -\frac{2496}{10625} & \frac{7488}{4375} \\ \frac{45864}{2125} & \frac{307633}{180625} & -\frac{338688}{903125} & \frac{145152}{53125} \\ \frac{122304}{10625} & -\frac{338688}{903125} & \frac{5418793}{4515625} & -\frac{387072}{265625} \\ \frac{52416}{625} & \frac{145152}{53125} & -\frac{387072}{265625} & \frac{181513}{15625} \end{bmatrix}$$



```
(%i43) lzz[1].lzz[2];  
      [ 1  
      (%o43)  0  
              0  
              0
```

Qui sono definite anche le due funzioni ausiliarie  $v3norma(vv)$  e  $incart(ra,f,h)$ .

Le formule qui applicate sembrano banali, data la loro semplicita' ma... a me sembra che siano raramente usate in letteratura e nell'insegnamento...